The Main Bound Computations $\left(L \geq 27 + \frac{1}{50}\right)$

Here we assemble bounds on the relevant pieces and combine them to accomplish the bound in the Main Theorem.

Definitions

In[1]:=

```
<< IntervalTools`
```

In[2]:=

```
{Lmin, Lmax} = {27 + 1/50, ∞};
Linterval = Interval[{Lmin, Lmax}];
η[L_] = 18/(10 + 9 L);
c[L_] = 1 + 391/(683 + 74 L);
r[L_] = 149/140 + 769/(512 + 30 L);
σ1[L_] = c[L] + (c[L] - 1/2)^2/r[L];
δ[L_] = 2 c[L] - σ1[L] - 1/2;
θ-η[L_] = ArcCos[(-η[L] - c[L])/r[L]];
θ1+η[L_] = ArcCos[(1 + η[L] - c[L])/r[L]];
θ1-c[L_] = ArcCos[(1 - 2 c[L])/r[L]];
```

In[12]:=
```
Reduce[D[η[L], L] ≤ 0 && L ≥ 27, L, Reals]
Reduce[D[c[L], L] ≤ 0 && L ≥ 27, L, Reals]
Reduce[D[r[L], L] ≤ 0 && L ≥ 27, L, Reals]
Reduce[D[σ1[L], L] ≤ 0 && L ≥ 27, L, Reals]
Reduce[D[δ[L], L] ≤ 0 && L ≥ 27, L, Reals]
Reduce[D[θ-η[L], L] ≥ 0 && L ≥ 27, L, Reals]
Reduce[D[θ1+η[L], L] ≤  0 && L ≥ 27, L, Reals]
Reduce[D[θ1-c[L], L] ≥ 0 && L ≥ 27, L, Reals]
```

Out[12]= L ≥ 27

Out[13]= L ≥ 27

Out[14]= L ≥ 27

Out[15]= L ≥ 27

Out[16]= L ≥ 27

Out[17]= L ≥ 27

Out[18]= L ≥ 27

Out[19]= L ≥ 27

Thus, the following interval enclosures are justified. Actually, η, c, r only contain one occurrence of L, and so are automatically sharp.

In[20]:=
```
η[Lint_Interval] := Block[{L}, MonotoneEnclosure[η[L], L, Lint]];
c[Lint_Interval] := Block[{L}, MonotoneEnclosure[c[L], L, Lint]];
r[Lint_Interval] := Block[{L}, MonotoneEnclosure[r[L], L, Lint]];
σ1[Lint_Interval] := Block[{L}, MonotoneEnclosure[σ1[L], L, Lint]];
δ[Lint_Interval] := Block[{L}, MonotoneEnclosure[δ[L], L, Lint]];
θ-η[Lint_Interval] := Block[{L}, MonotoneEnclosure[θ-η[L], L, Lint]];
θ1+η[Lint_Interval] := Block[{L}, MonotoneEnclosure[θ1+η[L], L, Lint]];
θ1-c[Lint_Interval] := Block[{L}, MonotoneEnclosure[θ1-c[L], L, Lint]];
```

Additionally we will frequently encounter $\log\left(\frac{r}{c-1/2}\right)$, which is monotone decreasing:

In[28]:=
```
Reduce[D[Log[r[L] / (c[L] - 1/2)], L] ≤ 0 && L ≥ 27, L, Reals]
```

Out[28]= L ≥ 27

In[29]:=
```
logrc[L_] = Log[r[L] / (c[L] - 1/2)];
logrc[Lint_Interval] := Block[{L}, MonotoneEnclosure[Log[r[L] / (c[L] - 1/2)], L, Lint]]
```

In[31]:=
```
Er[a_, d_, T_] =
  4 (4 + 3 π)/(45 ((17 + 2 a)^2 + 4 T^2)^(3/2)) - 4 T/(3 ((17 + 2 a)^2 + 4 T^2)) + (8 + 6 π)/(45 ((17 + 2 a - 2 d)^2 + 4 T^2)^(3/2)) +
   2 T/(3 ((17 + 2 a - 2 d)^2 + 4 T^2)) + (8 + 6 π)/(45 ((17 + 2 a + 2 d)^2 + 4 T^2)^(3/2)) + 2 T/(3 ((17 + 2 a + 2 d)^2 + 4 T^2)) +
   2 ArcTan[(1 + 2 a)/(2 T)] + 2 ArcTan[(5 + 2 a)/(2 T)] + 2 ArcTan[(9 + 2 a)/(2 T)] + 2 ArcTan[(13 + 2 a)/(2 T)] -
   1/2 (15 + 2 a) ArcTan[(17 + 2 a)/(2 T)] - ArcTan[(1 + 2 a - 2 d)/(2 T)] - ArcTan[(5 + 2 a - 2 d)/(2 T)] -
   ArcTan[(9 + 2 a - 2 d)/(2 T)] - ArcTan[(13 + 2 a - 2 d)/(2 T)] + 1/4 (15 + 2 a - 2 d) ArcTan[(17 + 2 a - 2 d)/(2 T)] -
   ArcTan[(1 + 2 a + 2 d)/(2 T)] - ArcTan[(5 + 2 a + 2 d)/(2 T)] - ArcTan[(9 + 2 a + 2 d)/(2 T)] -
   ArcTan[(13 + 2 a + 2 d)/(2 T)] + 1/4 (15 + 2 a + 2 d) ArcTan[(17 + 2 a + 2 d)/(2 T)] +
   1/2 T Log[1 + (17 + 2 a)^2/(4 T^2)] - 1/4 T Log[1 + (17 + 2 a - 2 d)^2/(4 T^2)] - 1/4 T Log[1 + (17 + 2 a + 2 d)^2/(4 T^2)];
```

Bounds on the pieces

Domains for $L, \eta, c, r, \sigma_1, \delta$

In[32]:=
```
ηinterval = η[Linterval]
cinterval = c[Linterval]
rinterval = r[Linterval]
N[{ηinterval, cinterval, rinterval}, 3]
```

Out[32]= Interval[{0, 900/12659}]

Out[33]= Interval[{1, 76837/67062}]

Out[34]= Interval[{149/140, 1523637/925820}]

Out[35]= {Interval[{0, 0.0713}], Interval[{0.998, 1.15}], Interval[{1.06, 1.65}]}

```
In[36]:= σ1interval = σ1[Linterval]
         δinterval = δ[Linterval]
         N[{σ1interval, δinterval}, 3]
```

Out[36]= $\text{Interval}\left[\left\{\frac{184}{149}, \frac{4793676918843499}{3426135363028314}\right\}\right]$

Out[37]= $\text{Interval}\left[\left\{\frac{79}{298}, \frac{672158744063911}{1713067681514157}\right\}\right]$

Out[38]= {Interval[{1.23, 1.40}], Interval[{0.264, 0.393}]}

Inequalities:
$-\frac{1}{2} \leq c[L] - r[L] \leq 1 - c[L] \leq -\eta[L] < 0 < 1 < 1 + \eta[L] < c[L] < \sigma_1[L] < c[L] + r[L]$

In[39]:= $-\frac{1}{2} < c[L] - r[L] < 1 - c[L] < -\eta[L] < 0 < 1 < 1 + \eta[L] < c[L] < \sigma_1[L] < c[L] + r[L]$

Out[39]= $-\frac{1}{2} < -\frac{9}{140} - \frac{769}{512 + 30\,L} + \frac{391}{683 + 74\,L} < -\frac{391}{683 + 74\,L} < -\frac{18}{10 + 9\,L} < 0 < 1 < 1 + \frac{18}{10 + 9\,L} < 1 + \frac{391}{683 + 74\,L} < 1 + \frac{391}{683 + 74\,L} + \frac{\left(\frac{1}{2} + \frac{391}{683+74\,L}\right)^2}{\frac{149}{140} + \frac{769}{512+30\,L}} < \frac{289}{140} + \frac{769}{512 + 30\,L} + \frac{391}{683 + 74\,L}$

In[40]:= $N\left[\text{Reduce}\left[-\frac{1}{2} < c[L] - r[L] < 1 - c[L] < -\eta[L] < 0 < 1 < 1 + \eta[L] < c[L] < \sigma_1[L] < c[L] + r[L], L\right], 30\right]$

$\text{Reduce}\left[-\frac{1}{2} < c[L] - r[L] < 1 - c[L] < -\eta[L] < 0 < 1 < 1 + \eta[L] < c[L] < \sigma_1[L] < c[L] + r[L] \,\&\&\, L \geq \frac{1351}{50}, L\right]$

Out[40]= L > 27.0151561678270966757452912307

Out[41]= $L \geq \frac{1351}{50}$

Bound: $|g(a, T)| \leq \frac{2-a}{50\,T}$

In another file.

Bound: $\frac{E_\delta}{\pi} \leq \frac{(640+216\,a)\,d-112-39\,a}{1536\,(3\,T+3\,a-1)} + \frac{1}{2^{10}}$

In another file.

Bound 1: $|g(a, T)| + \frac{E_\delta}{\pi} \leq \frac{1}{2^{10}} + \frac{1/14}{T-1/5}$

In[42]:= $\text{bound}[1] = \frac{1}{2^{10}} + \frac{1/14}{T - 1/5};$

In[43]:=
```
Clear[f];
f[a_, d_, T_] = bound[1] - ((2 - a)/(50 T) + ((640 + 216 a) d - 112 - 39 a)/(1536 (3 T + 3 a - 1)) + 1/2^10);
(* for each a, the function f is monotone in d, and rational in T *)
f[a_, dint_Interval, Tint_Interval] :=
  IntervalHull[f[a, Min[dint], Tint], f[a, Max[dint], Tint]];
f[a_, d_, Tint_Interval] := Module[{T, tmin = Min[Tint], tmax = Max[Tint], min, max},
   min = First[Minimize[{f[a, d, T], tmin ≤ T ≤ tmax}, T, Reals]];
   max = First[Maximize[{f[a, d, T], tmin ≤ T ≤ tmax}, T, Reals]];
   Interval[{min, max}]];
```

In[47]:=
```
δinterval
```

Out[47]= Interval$\left[\left\{\frac{79}{298}, \frac{672\,158\,744\,063\,911}{1\,713\,067\,681\,514\,157}\right\}\right]$

In[48]:=
```
f[0, δinterval, Interval[{5 / 7, ∞}]]
f[1, δinterval, Interval[{5 / 7, ∞}]]
```

••• Minimize: The minimum is not attained at any point satisfying the given constraints.

••• Minimize: The minimum is not attained at any point satisfying the given constraints.

Out[48]= Interval$\left[\left\{0, \frac{2\,147\,303}{42\,912\,000}\right\}\right]$

••• Minimize: The minimum is not attained at any point satisfying the given constraints.

••• Minimize: The minimum is not attained at any point satisfying the given constraints.

Out[49]= Interval$\left[\left\{0, \frac{246\,294\,287}{2\,488\,896\,000}\right\}\right]$

The warning messages are just acknowledging that the extreme values are happening as $T \to \infty$.

Bound 2: $\frac{2}{\pi}\text{Log}[\text{Zeta}[\sigma_1]] - \frac{\text{Log}[\text{Zeta}[2c]]}{\text{Log}[\frac{r}{c-1/2}]} \leq \frac{80\,807+17\,909\,L+178\,L^2}{4\,(164\,296+9637\,L+128\,L^2)}$

In[50]:=
```
bound[2] = (80 807 + 17 909 L + 178 L^2)/(4 (164 296 + 9637 L + 128 L^2));
```

In[51]:=
```
Clear[f];
f[L_] = bound[2] - (2/π Log[Zeta[σ1[L]]] - Log[Zeta[2 c[L]]]/Log[r[L]/(c[L] - 1/2)]);
```

In[53]:=
```
Reduce[D[bound[2]] == 0 && L ≥ 27, L, Reals]
```

Out[53]= False

In[54]:=
```
f[Lint_Interval] := Module[{rat, L},
   rat = MonotoneEnclosure[(80807 + 17909 L + 178 L^2)/(4 (164296 + 9637 L + 128 L^2)), L, Lint];
   rat - (2/π Log[Zeta[σ1[Lint]]] - Log[Zeta[2 c[Lint]]]/logrc[Lint])];
```

In[55]:=
```
N[f[Interval[{2^15, ∞}]], 30]
```

Out[55]=
```
Interval[
 {0.000500758693829279976294113034809, 0.00170260069729536642276347599026}]
```

In[56]:=
```
Timing[{t0, t1, t2} = ProveNonNegative[f, f, {Linterval}, MaxDepth → 20];]
Length /@ {t0, t1, t2}
```

Out[56]=
```
{5.57471, Null}
```

Out[57]=
```
{0, 0, 989}
```

Bound 3: $\frac{1/\pi}{\mathrm{Log}[r/(c-1/2)]}\kappa_1 L \leq \frac{238413}{2^{20}} L + \frac{80396+135589 L+798 L^2}{16\left(38735+3105 L+32 L^2\right)}$

In[58]:=
```
κ1[L_] = (θ-η[L] - θ1+η[L]) * (1 + η[L] - c[L])/2 -
   (π - θ-η[L]) (c[L] - 1/2) + r[L] (Sin[θ-η[L]] + Sin[θ1+η[L]])/2;
```

In[59]:=
```
bound[3] = 238413/2^20 L + (80396 + 135589 L + 798 L^2)/(16 (38735 + 3105 L + 32 L^2));
```

This is done in another notebook.

Bound 4: $\frac{\mathrm{Log}[\mathrm{Zeta}[c]]}{2\,\mathrm{Log}\left[\frac{r}{c-1/2}\right]}\left(\frac{3}{2}+\frac{1}{2 J_1}+\frac{1}{J_2}+\frac{\theta_{1+\eta}}{\pi}+\left(1-\frac{1}{J_2}\right)\frac{\theta_{1-c}}{\pi}-\frac{\theta_{-\eta}}{\pi}\right) \leq \frac{149201-214796 L-1135 L^2}{496726+75117 L+512 L^2}+\frac{1365}{2^{10}}\mathrm{Log}[1+L]+\frac{L}{2^{20}}$

In[60]:=
```
bound[4] = (149201 - 214796 L - 1135 L^2)/(496726 + 75117 L + 512 L^2) + 1365/2^10 Log[1 + L] + L/2^20;
```

In[61]:=
```
Clear[f];
f[L_] = Block[{J1 = 64, J2 = 24},
```

$$\text{bound}[4] - \frac{\text{Log}[\text{Zeta}[c[L]]]}{2\,\text{logrc}[L]}\left(\frac{3}{2} + \frac{1}{2\,\text{J1}} + \frac{1}{\text{J2}} + \frac{\theta_{1+\eta}[L]}{\pi} + \left(1 - \frac{1}{\text{J2}}\right)\frac{\theta_{1-c}[L]}{\pi} - \frac{\theta_{-\eta}[L]}{\pi}\right)]$$

Out[62]=

$$\frac{L}{1048576} + \frac{149201 - 214796\,L - 1135\,L^2}{496726 + 75117\,L + 512\,L^2} + \frac{1365\,\text{Log}[1+L]}{1024} - \frac{\left(\frac{595}{384} - \frac{\text{ArcCos}\left[\frac{-1-\frac{18}{10+9L}-\frac{391}{683+74L}}{\frac{149}{140}+\frac{769}{512+30L}}\right]}{\pi} + \frac{\text{ArcCos}\left[\frac{\frac{18}{10+9L}-\frac{391}{683+74L}}{\frac{149}{140}+\frac{769}{512+30L}}\right]}{\pi} + \frac{23\,\text{ArcCos}\left[\frac{1-2\left(1+\frac{391}{683+74L}\right)}{\frac{149}{140}+\frac{769}{512+30L}}\right]}{24\,\pi}\right)\text{Log}\left[\text{Zeta}\left[1+\frac{391}{683+74L}\right]\right]}{2\,\text{Log}\left[\frac{\frac{149}{140}+\frac{769}{512+30L}}{\frac{1}{2}+\frac{391}{683+74L}}\right]}$$

In[63]:=
```
Plot[f[L], {L, 27.02, 100}, PlotRange → All]
```

Out[63]=



In[64]:=
```
logzetapluslog[1] = 0;
logzetapluslog[s_] = Log[Zeta[s]] + Log[s - 1];
logzetapluslog[sint_Interval] :=
  Block[{s}, MonotoneEnclosure[logzetapluslog[s], s, sint]];
```

In[67]:=

$$\text{logoverlog}[s\_, x\_] = \frac{\text{Log}[s+x]}{\text{Log}[1+x]};$$

```
logoverlog[s_, xint_Interval] :=
  Block[{x}, MonotoneEnclosure[logoverlog[s, x], x, xint]];
```

In[69]:=
```
FullSimplify[f[L] == L/2^20 + (149201 - 214796 L - 1135 L^2)/(496726 + 75117 L + 512 L^2) -
   (595/384 - θ_{-η}[L]/π + θ_{1+η}[L]/π + 23 θ_{1-c}[L]/(24 π))/(2 logrc[L]) (logzetapluslog[1 + 391/(683 + 74 L)]) +
   ((595/384 - θ_{-η}[L]/π + θ_{1+η}[L]/π + 23 θ_{1-c}[L]/(24 π))/(2 logrc[L]) Log[391/(683+74 L)]/Log[1 + L] + 1365/1024) Log[1 + L],
 Assumptions → L ≥ 27 + 1/50]
FullSimplify[Log[391/(683+74 L)]/Log[1 + L] == -Log[74/391]/Log[1 + L] - logoverlog[683/74, L],
 Assumptions → L ≥ 27 + 1/50]
```

Out[69]= True

Out[70]= True

In[71]:=
```
Reduce[D[L/1048576 + (149201 - 214796 L - 1135 L^2)/(496726 + 75117 L + 512 L^2), L] ≤ 0 && L ≥ Lmin, L, Reals]
Reduce[D[Log[391/(683+74 L)]/Log[1 + L], L] ≤ 0 && L ≥ 1351/50, L, Reals]
```

Out[71]= $\frac{1351}{50} \leq L \leq$ 99.6…

Out[72]= $L \geq \frac{1351}{50}$

In[73]:=
```
f[Lint_Interval] := Module[{L, firstterms, coeff},
   firstterms = PiecewiseMonotoneEnclosure[
     -1135/512 + L/1048576 + (640174922 - 24717757 L)/(512 (496726 + 75117 L + 512 L^2)), L, {99.6…}, Lint];

   coeff = (595/384 - θ_{-η}[Lint]/π + θ_{1+η}[Lint]/π + 23 θ_{1-c}[Lint]/(24 π)) / (2 logrc[Lint]);
   firstterms - coeff (logzetapluslog[1 + 391/(683 + 74 Lint)]) +
    (coeff (-Log[74/391]/Log[1 + Lint] - logoverlog[683/74, Lint]) + 1365/1024) Log[1 + Lint]];
```

In[74]:=
```
Plot[f[L], {L, 270.02, 270.04}]
f[Interval[{270 + 1/50, 270 + 2/50}]] // N
```

Out[74]=



Out[75]= Interval[{0.00831001, 0.00841002}]

In[76]:=
```
Timing[{fails, undec, proven} = ProveNonNegative[f, f, {Linterval}, MaxDepth → 20];]
{fails, undec, Length[proven]}
```

Out[76]= {17.1176, Null}

Out[77]= {{}, {}, 797}

Bound 5:

$$\frac{\text{Log}[\text{Zeta}[1+\eta]]}{2\,\text{Log}[\frac{r}{c-1/2}]}\left(-\frac{\theta_{1+\eta}}{\pi}+\frac{\theta_{1-c}}{\pi}+\frac{\theta_{-\eta}}{\pi}-\frac{1}{2}\right)\le\frac{79\,045-118\,430\,L-182\,L^2}{599\,789+91\,562\,L+512\,L^2}+\frac{529}{2^{10}}\text{Log}[1+L]+\frac{L}{2^{22}}$$

In[78]:=

$$\text{bound}[5]=\frac{79\,045-118\,430\,L-182\,L^2}{599\,789+91\,562\,L+512\,L^2}+\frac{529}{2^{10}}\text{Log}[1+L]+\frac{L}{2^{22}};$$

In[79]:=
```
Clear[f];
```

$$\text{f}[\text{L\_}]=\text{bound}[5]-\frac{\text{Log}[\text{Zeta}[1+\eta[L]]]}{2\,\text{logrc}[L]}\left(-\frac{\theta_{1+\eta}[L]}{\pi}+\frac{\theta_{1-c}[L]}{\pi}+\frac{\theta_{-\eta}[L]}{\pi}-\frac{1}{2}\right)$$

Out[80]=

$$\frac{L}{4\,194\,304}+\frac{79\,045-118\,430\,L-182\,L^2}{599\,789+91\,562\,L+512\,L^2}+\frac{529\,\text{Log}[1+L]}{1024}-\frac{\left(-\frac{1}{2}+\frac{\text{ArcCos}\left[\frac{-1-\frac{18}{10+9L}-\frac{391}{683+74L}}{\frac{149}{140}+\frac{769}{512+30L}}\right]}{\pi}-\frac{\text{ArcCos}\left[\frac{\frac{18}{10+9L}-\frac{391}{683+74L}}{\frac{149}{140}+\frac{769}{512+30L}}\right]}{\pi}+\frac{\text{ArcCos}\left[\frac{1-2\left(1+\frac{391}{683+74L}\right)}{\frac{149}{140}+\frac{769}{512+30L}}\right]}{\pi}\right)\text{Log}\left[\text{Zeta}\left[1+\frac{18}{10+9L}\right]\right]}{2\,\text{Log}\left[\frac{\frac{149}{140}+\frac{769}{512+30L}}{\frac{1}{2}+\frac{391}{683+74L}}\right]}$$

In[81]:=
```
Plot[f[L], {L, 27.02, 40}, PlotRange → All]
```

Out[81]=



In[82]:=
```
logzetapluslog[0] = 0;
logzetapluslog[s_] = Log[Zeta[s]] + Log[s - 1];
logzetapluslog[sint_Interval] :=
  Block[{s}, MonotoneEnclosure[logzetapluslog[s], s, sint]];
```

In[85]:=
```
logoverlog[s_, x_] = Log[s + x] / Log[1 + x];
logoverlog[s_, xint_Interval] :=
  Block[{x}, MonotoneEnclosure[logoverlog[s, x], x, xint]];
```

In[87]:=
```
FullSimplify[f[L] == L / 4 194 304 + (79 045 - 118 430 L - 182 L^2) / (599 789 + 91 562 L + 512 L^2) -
   ((- θ_{1+η}[L] / π + θ_{1-c}[L] / π + θ_{-η}[L] / π - 1/2) / (2 logrc[L])) (logzetapluslog[1 + η[L]]) +
   (((- θ_{1+η}[L] / π + θ_{1-c}[L] / π + θ_{-η}[L] / π - 1/2) / (2 logrc[L])) Log[η[L]] / Log[1 + L] + 529 / 1024) Log[1 + L],
  Assumptions → L ≥ 27 + 1 / 50]
FullSimplify[Log[18 / (10 + 9 L)] / Log[1 + L] == - Log[1 / 2] / Log[1 + L] - logoverlog[10 / 9, L],
  Assumptions → L ≥ 27 + 1 / 50]
```

Out[87]= True

Out[88]= True

In[89]:=
```
Reduce[D[L / 4 194 304 + (79 045 - 118 430 L - 182 L^2) / (599 789 + 91 562 L + 512 L^2), L] ≤ 0 && L ≥ Lmin, L, Reals]
```

Out[89]= $\frac{1351}{50} \le L \le$ 45.7…

In[90]:=
```
f[Lint_Interval] := Module[{L, firstterms, coeff},
   firstterms =
    PiecewiseMonotoneEnclosure[
     L/4194304 + (79045 - 118430 L - 182 L^2)/(599789 + 91562 L + 512 L^2), L, {45.7…}, Lint];
   coeff = (-θ_{1+η}[Lint]/π + θ_{1-c}[Lint]/π + θ_{-η}[Lint]/π - 1/2)/(2 logrc[Lint]);
   firstterms - coeff (logzetapluslog[1 + η[Lint]]) +
    (coeff (-Log[1/2]/Log[1 + Lint] - logoverlog[10/9, Lint]) + 529/1024) Log[1 + Lint]];
```

In[91]:=
```
Plot[f[L], {L, 27.02, 27.04}]
f[Interval[{27 + 1/50, 27 + 2/50}]] // N
```

Out[91]=



Out[92]= `Interval[{0.00640361, 0.00683416}]`

In[93]:= `N[f[Interval[{2^22, ∞}]], 30]`

Out[93]= `Interval[{0.674102896507041009666626751466, ∞}]`

In[94]:=
```
Timing[{fails, undec, proven} = ProveNonNegative[f, f, {Linterval}, MaxDepth → 25];]
{fails, undec, Length[proven]}
```

Out[94]= `{9.29132, Null}`

Out[95]= `{{}, {}, 172}`

Bound 6: $\kappa_2 = \frac{\pi}{4 J_1}\left(\mathrm{Log}[\mathrm{Zeta}[c+r]] + 2\,\mathrm{Sum}\left[\mathrm{Log}\left[\mathrm{Zeta}\left[c + r\,\mathrm{Cos}\left[\frac{\pi j}{2 J_1}\right]\right]\right], \{j, 1, J_1 - 1\}\right]\right)$

$$\frac{\kappa_2/\pi}{\mathrm{Log}[r/(c-1/2)]} \leq \frac{635}{1024} - \frac{9\,(25384532+113745\,L)}{64\,(7149852+150141\,L+512\,L^2)}$$

In[96]:=
```
bound[6] = 635/1024 - 9 (25384532 + 113745 L)/(64 (7149852 + 150141 L + 512 L^2));
```

In[97]:=
```
Clear[f];
f[L_] = bound[6] - Block[{J1 = 64}, (1/π)/logrc[L] π/(4 J1)
    (Log[Zeta[c[L] + r[L]]] + 2 Sum[Log[Zeta[c[L] + r[L] Cos[π j/(2 J1)]]], {j, 1, J1 - 1}])];
```

In[99]:=
```
Reduce[D[bound[6], L] ≥ 0 && L ≥ 27]
```

Out[99]=
```
L ≥ 27
```

In[100]:=
```
f[Lint_Interval] := Module[{L, rat, J1 = 64},
   rat = MonotoneEnclosure[635/1024 - 9 (25 384 532 + 113 745 L)/(64 (7 149 852 + 150 141 L + 512 L^2)), L, Lint];
   rat - (1/π)/logrc[Lint] π/(4 J1) (Log[Zeta[c[Lint] + r[Lint]]] +
       2 Sum[Log[Zeta[c[Lint] + r[Lint] Cos[π j/(2 J1)]]], {j, 1, J1 - 1}])];
```

In[101]:=
```
N[f[Interval[{2^16, ∞}]], 30]
```

Out[101]=
```
Interval[{0.00190116149192584718501629953234, 0.00284634858015494862148462702167}]
```

In[102]:=
```
N[f[Interval[{27 + 1/50, 28}]], 30]
```

Out[102]=
```
Interval[
  {-0.00277775647380329995821780181877, 0.00482659079256835664310778475669}]
```

In[103]:=
```
Timing[Length /@ ProveNonNegative[f, f, {Linterval}, MaxDepth → 20]]
```

Out[103]=
```
{88.3238, {0, 0, 369}}
```

Bound 7: $\kappa_3 = \frac{\pi-\theta_{1-c}}{2 J_2}\left(\text{Log[Zeta}[1-c+r]] + 2\,\text{Sum}\left[\text{Log}\left[\text{Zeta}\left[1-c-r\,\text{Cos}\left[\theta_{1-c}+(\pi-\theta_{1-c})\,\frac{j}{J_2}\right]\right]\right], \{j, 1, J_2-1\}\right]\right)$

$$\frac{\kappa_3/\pi}{\text{Log}[r/(c-1/2)]} \le \frac{491}{1024} - \frac{33\,179\,656+3\,346\,893\,L}{512\,(208\,616+21\,113\,L+512\,L^2)}$$

In[104]:=
```
bound[7] = 491/1024 - (33 179 656 + 3 346 893 L)/(512 (208 616 + 21 113 L + 512 L^2));
```

In[105]:=
```
Clear[f];
f[L_] = 491/1024 - (33 179 656 + 3 346 893 L)/(512 (208 616 + 21 113 L + 512 L^2)) -
   Block[{J2 = 24}, (1/π)/logrc[L] (π - θ1-c[L])/(2 J2) (Log[Zeta[1 - c[L] + r[L]]] +
       2 Sum[Log[Zeta[1 - c[L] - r[L] Cos[θ1-c[L] (1 - j/J2) + π j/J2]]], {j, 1, J2 - 1}])];
```

```
In[107]:= Reduce[D[bound[7], L] ≥ 0 && L ≥ 27]
          Reduce[D[1 - c[L] + r[L], L] ≤ 0 && L ≥ 27]
Out[107]= L ≥ 27
Out[108]= L ≥ 27

In[109]:= f[Lint_] := Module[{L, rat, J2, rmcp1, insidezeta},
            J2 = 24;
            rat = MonotoneEnclosure[491/1024 - (33179656 + 3346893 L)/(512 (208616 + 21113 L + 512 L^2)), L, Lint];
            rmcp1 = Log[Zeta[MonotoneEnclosure[1 - c[L] + r[L], L, Lint]]];
            insidezeta[j_] := IntervalIntersection[Interval[{Min[c[Lint]], ∞}],
              1 - c[Lint] - r[Lint] * Cos[θ1-c[Lint] * (1 - j/J2) + π j/J2]];
            rat - (1/π)/logrc[Lint] (π - θ1-c[Lint])/(2 J2)
              (rmcp1 + 2 Sum[Log[Zeta[insidezeta[j]]], {j, 1, J2 - 1}])];

In[110]:= N[f[Interval[{27 + 1/50, 27 + 1/50 + 1/100}]], 30]
          N[f[Interval[{2^20, ∞}]], 30]
Out[110]= Interval[{0.00151497267607420503100595273910, 0.00164915403486988469788854466171}]
Out[111]= Interval[{0.00267492113348354076383606546959, 0.00292377330749874296642252703956}]

In[112]:= Timing[Length /@ ProveNonNegative[f, f, {Linterval}, MaxDepth → 25]]
Out[112]= {141.589, {0, 0, 473}}
```

Bound 8: $\frac{\kappa_7/\pi}{\mathrm{Log}[r/(c-1/2)]} \leq \frac{18281-1411\,L-50\,L^2}{800695+63962\,L+512\,L^2}$

```
In[113]:= bound[8] = (18281 - 1411 L - 50 L^2)/(800695 + 63962 L + 512 L^2);
```

The quantity κ_7 is defined by

```
In[114]:= Clear[K7];
```

In[115]:=

$$K7[c\_, r\_, \eta\_] = \frac{1}{4} \text{Block}\Big[\{\sigma = c + r \text{Cos}[\theta], t = r \text{Sin}[\theta]\}, \text{Integrate}\Big[(1+\eta-\sigma)\left(2t - 4 + \frac{7}{19}\left((1+\sigma)^2 + (t-2)^2\right)\right), \left\{\theta, \text{ArcCos}\left[\frac{1+\eta-c}{r}\right], \text{ArcCos}\left[\frac{-\eta-c}{r}\right]\right\}\Big]\Big]$$

Out[115]=

$$\frac{1}{152}\Bigg(20(1-c+\eta)^2 - 20(-1+c-\eta)(c+\eta) - 2r\left(-55+14c^2+7r^2+c(7-21\eta)-21\eta\right)\sqrt{-\frac{c^2-r^2+2c\eta+\eta^2}{r^2}} + 2r\left(-48+14c^2+7r^2-7c(-2+\eta)-7\eta\right)\sqrt{1-\frac{(1-c+\eta)^2}{r^2}} + 2\left(41+7c^3+c\left(-55+14r^2-14\eta\right)-7c^2(-1+\eta)+\left(41-7r^2\right)\eta\right)\text{ArcCos}\left[\frac{1-c+\eta}{r}\right] - 2\left(41+7c^3+c\left(-55+14r^2-14\eta\right)-7c^2(-1+\eta)+\left(41-7r^2\right)\eta\right)\text{ArcCos}\left[-\frac{c+\eta}{r}\right] - 5r^2\,\text{Cos}\left[2\,\text{ArcCos}\left[\frac{1-c+\eta}{r}\right]\right] + 5r^2\,\text{Cos}\left[2\,\text{ArcCos}\left[-\frac{c+\eta}{r}\right]\right]\Bigg)$$

Some nontrivial rearranging work, which we did by hand, gives

In[116]:=

$$\kappa_7[c\_, r\_, \eta\_] := \text{Block}\Big[\Big\{\alpha1 = \sqrt{(-1+c+r-\eta)(1-c+r+\eta)}, \alpha2 = \sqrt{-(c-r+\eta)(c+r+\eta)}\Big\}, \frac{1}{76}\Big(\alpha1\left(-48+14c+14c^2+7r^2-7\eta-7c\eta\right) + \alpha2\left(55-7c-14c^2-7r^2+21\eta+21c\eta\right) + 5(1+2\eta)^2 + \left(41-55c+7c^2+7c^3+14cr^2+41\eta-14c\eta-7c^2\eta-7r^2\eta\right)\left(\text{ArcTan}\left[\frac{-1+c-\eta}{\alpha1}\right] - \text{ArcTan}\left[\frac{c+\eta}{\alpha2}\right]\right)\Big)\Big]$$

Mathematica is unable to ascertain if the two expressions are equivalent, but they are.

In[117]:=

$$\text{(* Should Simplify to just 0 *)}\text{FullSimplify}\Big[K7[c, r, \eta] - \kappa_7[c, r, \eta], \text{Assumptions} \to \frac{-1}{2} < c - r < 1 - c < -\eta < 0 < 1 < 1 + \eta < c < c + r\Big]$$

Out[117]=

$$-\frac{1}{152}\left(c\left(-55+7c(1+c)+14r^2\right) - 7\left(c(2+c)+r^2\right)\eta + 41(1+\eta)\right)\left(\pi + 2 i\,\text{ArcCosh}\left[\frac{1-c+\eta}{r}\right] + 2\,\text{ArcSin}\left[\frac{c+\eta}{r}\right] + 2\,\text{ArcTan}\left[\frac{-1+c-\eta}{\sqrt{(-1+c+r-\eta)(1-c+r+\eta)}}\right] - 2 i\,\text{ArcTanh}\left[\frac{c+\eta}{\sqrt{(c-r+\eta)(c+r+\eta)}}\right]\right)$$

In[118]:=
```
(* Just as a reality check, compute the difference
 for 10 random numbers to 100000 digits. Should all be 0. *)
$MaxExtraPrecision = 100 000;
MinMax[Table[Block[{k = RandomInteger[{28, 10 000}]},
   N[K7[c[k], r[k], η[k]] - κ7[c[k], r[k], η[k]], 20]], {10}]]
$MaxExtraPrecision = 50;
```

N: Internal precision limit $MaxExtraPrecision = 100000.` reached while evaluating

$$\frac{1}{76}\left(-\frac{56180}{9409}-\frac{32953838330272431\sqrt{\frac{1442561450135187}{2}}}{5556167494669360 9353625}+\frac{323878868495277\sqrt{\frac{126071479227330907}{2}}}{11112334989338721870 7250}-\frac{9744796469069435269\left(-\text{ArcTan}[\text{Times}[\ll 2\gg]]+\text{ArcTan}\left[\frac{38064775}{2}\text{Power}[\ll 2\gg]\right]\right)}{295543771762051400}\right)+\frac{1}{152}\left(-\frac{2602300}{2098207}+\ll 9\gg\right).$$

N: Internal precision limit $MaxExtraPrecision = 100000.` reached while evaluating

$$\frac{1}{76}\left(-\frac{141158161}{28060805}-\frac{13340840656209809977773951\sqrt{\frac{280538914893910902913816189}{5}}}{962917066626617997280594335635304 1600}+\frac{603848304898864817018 9871\sqrt{\frac{2\ll 25\gg 89}{5}}}{962917066626617997280594335635304 1600}-\frac{2090260394550423334267223053143\left(-\text{ArcTan}[\text{Times}[\ll 2\gg]]+\text{ArcTan}[229183748404\,\text{Power}[\ll 2\gg]]\right)}{1271412921721892125717 76759000}\right)+\frac{1}{152}\left(-\frac{1\ll 9\gg 04}{2\ll 10\gg 65}+\ll 9\gg\right).$$

N: Internal precision limit $MaxExtraPrecision = 100000.` reached while evaluating

$$\frac{1}{76}\left(-\frac{1058658005}{211469764}-\frac{586192396985797998218270317 11\sqrt{962073466399214968285895675159}}{593469532447525008383537395319846486 3626000}+\frac{272911744870763668768090539\sqrt{\ll 33\gg}}{5395177567704772\ll 10\gg 9381678623966000}-\left(210850500273345799984366843585667 93\left(-\text{ArcTan}[\text{Times}[\ll 2\gg]]+\text{ArcTan}[12010740313775\,\text{Power}[\ll 2\gg]]\right)\right)/1319760788376407700288311920258300\right)+\frac{1}{152}(\ll 1\gg).$$

General: Further output of N::meprec will be suppressed during this calculation.

Out[119]= $\{0.\times 10^{-100\,020}, 0.\times 10^{-100\,020}\}$

Now we turn to developing an interval enclosure for κ_7.

```
In[121]:= (* the alphas are monotone decreasing*)
Reduce[L ≥ 27 && D[(-1 + c[L] + r[L] - η[L]) (1 - c[L] + r[L] + η[L]), L] ≤ 0, L, Reals]
Reduce[L ≥ 27 && D[-(c[L] - r[L] + η[L]) (c[L] + r[L] + η[L]), L] ≤ 0, L, Reals]

Out[121]= L ≥ 27

Out[122]= L ≥ 27

In[123]:= (* the alpha coefficients are monotone, one decreasing and one increasing *)
Reduce[
 L ≥ 27 && D[(-48 + 14 c[L] + 14 c[L]^2 + 7 r[L]^2 - 7 η[L] - 7 c[L] × η[L]), L] ≤ 0, L, Reals]
Reduce[L ≥ 27 && D[(55 - 7 c[L] - 14 c[L]^2 - 7 r[L]^2 + 21 η[L] + 21 c[L] × η[L]), L] ≥ 0,
 L, Reals]

Out[123]= L ≥ 27

Out[124]= L ≥ 27

In[125]:= (*the arctan coefficient is monotone decreasing *)
Reduce[L ≥ 27 && D[(41 - 55 c[L] + 7 c[L]^2 + 7 c[L]^3 + 14 c[L] r[L]^2 +
     41 η[L] - 14 c[L] × η[L] - 7 c[L]^2 η[L] - 7 r[L]^2 η[L]), L] ≤ 0, L, Reals]

Out[125]= L ≥ 27

In[126]:= kappa7[L_] = κ7[c[L], r[L], η[L]];
kappa7[Lint_Interval] := Module[{a1c, a2c, arctanc, L, α1, α2},
  α1 =
   Sqrt[MonotoneEnclosure[(-1 + c[L] + r[L] - η[L]) (1 - c[L] + r[L] + η[L]), L, Lint]];
  α2 = Sqrt[MonotoneEnclosure[-(c[L] - r[L] + η[L]) (c[L] + r[L] + η[L]), L, Lint]];
  a1c = MonotoneEnclosure[
    (-48 + 14 c[L] + 14 c[L]^2 + 7 r[L]^2 - 7 η[L] - 7 c[L] × η[L]), L, Lint];
  a2c = MonotoneEnclosure[(55 - 7 c[L] - 14 c[L]^2 - 7 r[L]^2 + 21 η[L] + 21 c[L] × η[L]),
    L, Lint];
  arctanc = MonotoneEnclosure[(41 - 55 c[L] + 7 c[L]^2 + 7 c[L]^3 + 14 c[L] r[L]^2 +
      41 η[L] - 14 c[L] × η[L] - 7 c[L]^2 η[L] - 7 r[L]^2 η[L]), L, Lint];

  1/76 (α1 * a1c + α2 * a2c + 5 (1 + 2 η[Lint])^2 +
     arctanc (ArcTan[(-1 + c[Lint] - η[Lint])/α1] - ArcTan[(c[Lint] + η[Lint])/α2]))]
```

In[128]:=
```
Plot[kappa7[L], {L, 27.02, 28}]
N[kappa7[Interval[{27 + 1 / 50, 28}]], 20]
```

Out[128]=



Out[129]=
```
Interval[{-0.074095309014146524890, -0.042662563846630312688}]
```

In[130]:=
```
Reduce[L ≥ 27 && D[bound[8], L] ≤ 0, L, Reals]
```

Out[130]=
```
L ≥ 27
```

In[131]:=
```
Clear[f];
f[L_] = bound[8] - (kappa7[L] / π)/(logrc[L]);
f[Lint_Interval] :=
  Module[{L}, MonotoneEnclosure[(18 281 - 1411 L - 50 L^2)/(800 695 + 63 962 L + 512 L^2), L, Lint] - (kappa7[Lint] / π)/(logrc[Lint])];
```

In[134]:=
```
N[f[Interval[{2^15, ∞}]], 30]
```

Out[134]=
```
Interval[
 {0.0000332388214824445357102907788924, 0.00112049541717487618317970614404}]
```

In[135]:=
```
Timing[Length /@ ProveNonNegative[f, f, {Linterval}, MaxDepth → 20]]
```

Out[135]=
```
{45.0998, {0, 0, 2213}}
```

Bound 9: $\dfrac{\kappa_8/\pi}{\mathrm{Log}[r/(c-1/2)]} \leq \dfrac{-961\,048-15\,293\,L-42\,L^2}{3\,255\,348+113\,665\,L+512\,L^2}$

In[136]:=
```
bound[9] = (- 961 048 - 15 293 L - 42 L^2)/(3 255 348 + 113 665 L + 512 L^2);
```

For the current values of c, r, we have $\theta_{-1/2} = \pi$. The quantity κ_8 is defined by

In[137]:=
```
K8 = 1/4 Block[{σ = c + r Cos[θ], t = r Sin[θ]},
   Integrate[(1 - 2 σ) (2 t - 4 + 7/19 ((1 + σ)^2 + (t - 2)^2)), {θ, ArcCos[(-η - c)/r], π}]]
```

Out[137]=
$$\frac{1}{76}\left(5\,r\,(2-4\,c+r)-\pi\,\left(41+7\,r^2+c\,\left(-96+7\,c\,(3+2\,c)+28\,r^2\right)\right)+\right.$$
$$2\,r\,\left(-48+14\,c^2+7\,r^2-7\,c\,(-2+\eta)-7\,\eta\right)\sqrt{-\frac{c^2-r^2+2\,c\,\eta+\eta^2}{r^2}}+$$
$$\left.5\,\left(-2\,c+2\,c^2+r^2-2\,\eta\,(1+\eta)\right)+\left(41+7\,r^2+c\,\left(-96+7\,c\,(3+2\,c)+28\,r^2\right)\right)\mathrm{ArcCos}\left[-\frac{c+\eta}{r}\right]\right)$$

Some easy rearranging work gives

In[138]:=
```
κ8[c_, r_, η_] :=
  Block[{α2 = Sqrt[-(c - r + η) (c + r + η)]}, 1/76 (c^2 (10 - 21 π) - 41 π - 14 c^3 π + 10 r^2 - 7 π r^2 +
     c (-10 + 96 π - 20 r - 28 π r^2) + 10 (r - η (1 + η)) + 2 (-48 + 7 r^2 + 7 c (2 + 2 c - η) - 7 η) α2 +
     (41 + 7 r^2 + c (-96 + 7 c (3 + 2 c) + 28 r^2)) ArcCos[-(c + η)/r])]
```

Mathematica is able to ascertain if the two expressions are equivalent, and they are.

In[139]:=
```
Reduce[K8 == κ8[c, r, η] && -1/2 < c - r < 1 - c < -η < 0 < 1 < 1 + η < c < c + r]
```

Out[139]=
$$\left(1<r\leq\frac{3}{2}\ \&\&\ 1<c<\frac{1+r}{2}\ \&\&\ 0<\eta<-1+c\right)\ ||$$
$$\left(\frac{3}{2}<r<2\ \&\&\ \frac{1}{2}(-1+2\,r)<c<\frac{1+r}{2}\ \&\&\ 0<\eta<-1+c\right)$$

Now we turn to developing an interval enclosure for κ_8.

In[140]:=
```
(* the α2 is monotone decreasing, and so is its coefficient *)
Reduce[L ≥ 27 && D[-(c[L] - r[L] + η[L]) (c[L] + r[L] + η[L]), L] ≤ 0, L, Reals]
Reduce[L ≥ 27 && D[2 (-48 + 7 r[L]^2 + 7 c[L] (2 + 2 c[L] - η[L]) - 7 η[L]), L] ≤ 0, L, Reals]
```

Out[140]= $L \geq 27$

Out[141]= $L \geq 27$

In[142]:=
```
(* the big terms are monotone increasing*)
Reduce[L ≥ 27 && D[c[L]^2 (10 - 21 π) - 41 π - 14 c[L]^3 π + 10 r[L]^2 - 7 π r[L]^2 +
     c[L] (-10 + 96 π - 20 r[L] - 28 π r[L]^2) + 10 (r[L] - η[L] (1 + η[L])), L] ≥ 0, L, Reals]
```

Out[142]= $L \geq 27$

In[143]:=
```
(*the arccos coefficient is monotone decreasing *)
Reduce[
 L ≥ 27 && D[(41 + 7 r[L]^2 + c[L] (-96 + 7 c[L] (3 + 2 c[L]) + 28 r[L]^2)), L] ≤ 0, L, Reals]
(* the arccos term is monotone *)
Reduce[L ≥ 27 && D[-(c[L] + η[L])/r[L], L] ≤ 0, L, Reals]
```

Out[143]= L ≥ 27

Out[144]= L ≥ 27

In[145]:=
```
kappa8[L_] = κ8[c[L], r[L], η[L]];
kappa8[Lint_Interval] := Module[{a2c, arccosc, arccos, L, bigs, α2},
  α2 = Sqrt[MonotoneEnclosure[-(c[L] - r[L] + η[L]) (c[L] + r[L] + η[L]), L, Lint]];
  a2c = MonotoneEnclosure[2 (-48 + 7 r[L]^2 + 7 c[L] (2 + 2 c[L] - η[L]) - 7 η[L]), L, Lint];
  arccos = ArcCos[MonotoneEnclosure[-(c[L] + η[L])/r[L], L, Lint]];
  arccosc = MonotoneEnclosure[
    (41 + 7 r[L]^2 + c[L] (-96 + 7 c[L] (3 + 2 c[L]) + 28 r[L]^2)), L, Lint];
  bigs = MonotoneEnclosure[c[L]^2 (10 - 21 π) - 41 π - 14 c[L]^3 π + 10 r[L]^2 - 7 π r[L]^2 +
     c[L] (-10 + 96 π - 20 r[L] - 28 π r[L]^2) + 10 (r[L] - η[L] (1 + η[L])), L, Lint];

  1/76 (bigs + a2c * α2 + arccosc * arccos)]
```

In[147]:=
```
Plot[kappa8[L], {L, 270.02, 280}]
N[kappa8[Interval[{270 + 1/50, 280}]], 20]
```

Out[147]=



Out[148]= Interval[{-0.29852563525562772615, -0.27528440912884977799}]

In[149]:=
```
Reduce[L ≥ 27 && D[bound[9], L] ≥ 0, L, Reals]
```

Out[149]= L ≥ 27

In[150]:=
```
Clear[f];
f[L_] = bound[9] - (kappa8[L] / π)/logrc[L];
f[Lint_Interval] := Module[{L},
   MonotoneEnclosure[(-961 048 - 15 293 L - 42 L^2)/(3 255 348 + 113 665 L + 512 L^2), L, Lint] - (kappa8[Lint] / π)/logrc[Lint]];
```

In[153]:= `N[f[Interval[{2^16, ∞}]], 30]`

Out[153]= `Interval[{0.000270211633843625302209032182015, 0.00168747585190771594561288523705}]`

In[154]:= `Timing[Length /@ ProveNonNegative[f, f, {Linterval}, MaxDepth → 20]]`

Out[154]= `{114.986, {0, 0, 10 327}}`

Assembling the final bound

In[155]:= `sharp[L_, T_] = Sum[bound[k], {k, 1, 7}] + 1/(T + 2) (bound[8] + bound[9])`

Out[155]=

$$\frac{1127}{1024}+\frac{953657\,L}{4194304}+\frac{80807+17909\,L+178\,L^2}{4\,(164296+9637\,L+128\,L^2)}-\frac{33179656+3346893\,L}{512\,(208616+21113\,L+512\,L^2)}+\frac{149201-214796\,L-1135\,L^2}{496726+75117\,L+512\,L^2}+\frac{79045-118430\,L-182\,L^2}{599789+91562\,L+512\,L^2}-\frac{9\,(25384532+113745\,L)}{64\,(7149852+150141\,L+512\,L^2)}+\frac{80396+135589\,L+798\,L^2}{16\,(38735+3105\,L+32\,L^2)}+\frac{1}{14\left(-\frac{1}{5}+T\right)}+\frac{\frac{18281-1411\,L-50\,L^2}{800695+63962\,L+512\,L^2}+\frac{-961048-15293\,L-42\,L^2}{3255348+113665\,L+512\,L^2}}{2+T}+\frac{947}{512}\,\mathrm{Log}[1+L]$$

First we wrestle with the terms involving T.

In[156]:=
```
Clear[f];
f[L_, T_] = 1/(14 (-1/5 + T)) + ((18 281-1411 L-50 L^2)/(800 695+63 962 L+512 L^2) + (-961 048-15 293 L-42 L^2)/(3 255 348+113 665 L+512 L^2))/(2 + T);
```

In[158]:= `Reduce[D[(18 281 - 1411 L - 50 L^2)/(800 695 + 63 962 L + 512 L^2) + (-961 048 - 15 293 L - 42 L^2)/(3 255 348 + 113 665 L + 512 L^2), L] ≥ 0 && L ≥ 27]`

Out[158]= `L ≥ 27`

That is, f is monotone increasing in L. It is not monotone in T, but no matter.

In[159]:=
```
f[Lint_Interval, Tint_Interval] := Module[{Lmin, Lmax, Aint},
  Aint =
   MonotoneEnclosure[(18281 - 1411 L - 50 L^2)/(800695 + 63962 L + 512 L^2) + (-961048 - 15293 L - 42 L^2)/(3255348 + 113665 L + 512 L^2), L, Lint];
  (1/14)/(Tint - 1/5) + Aint/(2 + Tint)]
```

In[160]:=
```
f[Linterval, Interval[{5/7, ∞}]]
```

Out[160]=
```
Interval[{-1740511359623068519/20621697391729192303, 5/36}]
```

In[161]:=
```
Maximize[{Limit[f[L, T], L → Infinity], T ≥ 5/7}, T, Reals]
```

Out[161]=
```
{1591/21888, {T → 5/7}}
```

In[162]:=
```
tobeprovennonneg[L_, T_] := 75/1024 - f[L, T];
```

In[163]:=
```
AbsoluteTiming[ProveNonNegative[tobeprovennonneg,
  tobeprovennonneg, {Linterval, Interval[{5/7, ∞}]}]]
```

Out[163]=
```
{0.020345,
 {{}, {}, {{Interval[{1351/50, 32}], Interval[{5/7, 1}]}, {Interval[{1351/50, 28}],
    Interval[{1, 2}]}, {Interval[{1351/50, 28}], Interval[{2, ∞}]},
   {Interval[{28, 32}], Interval[{1, 2}]}, {Interval[{28, 32}], Interval[{2, ∞}]},
   {Interval[{32, 64}], Interval[{5/7, 3/4}]}, {Interval[{32, 64}], Interval[{3/4, 1}]},
   {Interval[{64, 128}], Interval[{5/7, 23/32}]}, {Interval[{64, 128}],
    Interval[{23/32, 3/4}]}, {Interval[{128, ∞}], Interval[{5/7, 23/32}]},
   {Interval[{128, ∞}], Interval[{23/32, 3/4}]}, {Interval[{64, ∞}], Interval[{3/4, 1}]},
   {Interval[{32, 64}], Interval[{1, 2}]}, {Interval[{32, 64}], Interval[{2, ∞}]},
   {Interval[{64, ∞}], Interval[{1, 2}]}, {Interval[{64, ∞}], Interval[{2, ∞}]}}}}
```

In[164]:=
$$\text{sharp2}[L\_] = \frac{1127}{1024} + \frac{953\,657\,L}{4\,194\,304} + \frac{80\,807 + 17\,909\,L + 178\,L^2}{4\,(164\,296 + 9637\,L + 128\,L^2)} - \frac{33\,179\,656 + 3\,346\,893\,L}{512\,(208\,616 + 21\,113\,L + 512\,L^2)} + \frac{149\,201 - 214\,796\,L - 1135\,L^2}{496\,726 + 75\,117\,L + 512\,L^2} + \frac{79\,045 - 118\,430\,L - 182\,L^2}{599\,789 + 91\,562\,L + 512\,L^2} - \frac{9\,(25\,384\,532 + 113\,745\,L)}{64\,(7\,149\,852 + 150\,141\,L + 512\,L^2)} + \frac{80\,396 + 135\,589\,L + 798\,L^2}{16\,(38\,735 + 3105\,L + 32\,L^2)} + \frac{75}{1024} + \frac{947}{512}\,\text{Log}[1 + L]$$

Out[164]=
$$\frac{601}{512} + \frac{953\,657\,L}{4\,194\,304} + \frac{80\,807 + 17\,909\,L + 178\,L^2}{4\,(164\,296 + 9637\,L + 128\,L^2)} - \frac{33\,179\,656 + 3\,346\,893\,L}{512\,(208\,616 + 21\,113\,L + 512\,L^2)} + \frac{149\,201 - 214\,796\,L - 1135\,L^2}{496\,726 + 75\,117\,L + 512\,L^2} + \frac{79\,045 - 118\,430\,L - 182\,L^2}{599\,789 + 91\,562\,L + 512\,L^2} - \frac{9\,(25\,384\,532 + 113\,745\,L)}{64\,(7\,149\,852 + 150\,141\,L + 512\,L^2)} + \frac{80\,396 + 135\,589\,L + 798\,L^2}{16\,(38\,735 + 3105\,L + 32\,L^2)} + \frac{947}{512}\,\text{Log}[1 + L]$$

In[165]:=
$$N\left[\frac{953\,657}{4\,194\,304}, 20\right]$$

Out[165]= 0.22736954689025878906

In[166]:=
```
Plot[0.22737 L + 2 Log[1 + L] - 1/2 - sharp2[L], {L, 27, 20^3},
 PlotRange → All, AxesOrigin → {27.02, 0}, PlotLegends → "Expressions"]
```

Out[166]=


In[167]:=
$$\text{toprovenonneg}[L\_] = \frac{22\,737}{100\,000}\,L + 2\,\text{Log}[1 + L] - 1/2 - \text{sharp2}[L]$$

Out[167]=
$$-\frac{857}{512} + \frac{5939\,L}{13\,107\,200\,000} - \frac{80\,807 + 17\,909\,L + 178\,L^2}{4\,(164\,296 + 9637\,L + 128\,L^2)} + \frac{33\,179\,656 + 3\,346\,893\,L}{512\,(208\,616 + 21\,113\,L + 512\,L^2)} - \frac{149\,201 - 214\,796\,L - 1135\,L^2}{496\,726 + 75\,117\,L + 512\,L^2} - \frac{79\,045 - 118\,430\,L - 182\,L^2}{599\,789 + 91\,562\,L + 512\,L^2} + \frac{9\,(25\,384\,532 + 113\,745\,L)}{64\,(7\,149\,852 + 150\,141\,L + 512\,L^2)} - \frac{80\,396 + 135\,589\,L + 798\,L^2}{16\,(38\,735 + 3105\,L + 32\,L^2)} + \frac{77}{512}\,\text{Log}[1 + L]$$

Notice that all of the denominators are positive $L > 0$. The rational part of toprovenonneg[L] has only one critical value for $L \geq 27 + \frac{1}{50}$, near $L = 20\,000$. We use that knowledge to speed up the interval enclosure below, but don't need to do so.

In[168]:=
$$\text{Reduce}\Big[L \geq \text{Lmin} \,\&\&\, D\Big[-\frac{857}{512} - \frac{80807 + 17909 L + 178 L^2}{4 \left(164296 + 9637 L + 128 L^2\right)} + \frac{33179656 + 3346893 L}{512 \left(208616 + 21113 L + 512 L^2\right)} - \frac{149201 - 214796 L - 1135 L^2}{496726 + 75117 L + 512 L^2} - \frac{79045 - 118430 L - 182 L^2}{599789 + 91562 L + 512 L^2} + \frac{9 \left(25384532 + 113745 L\right)}{64 \left(7149852 + 150141 L + 512 L^2\right)} - \frac{80396 + 135589 L + 798 L^2}{16 \left(38735 + 3105 L + 32 L^2\right)}, L\Big] \leq 0\Big]$$

Out[168]=
$$L \geq \frac{1351}{50}$$

In[169]:=
$$\text{toprovenonneg[Lint\_Interval]} := \text{Module}\Big[\{L, \text{expr}, \text{exprint}\},$$
$$\text{expr} = -\frac{857}{512} - \frac{80807 + 17909 L + 178 L^2}{4 \left(164296 + 9637 L + 128 L^2\right)} + \frac{33179656 + 3346893 L}{512 \left(208616 + 21113 L + 512 L^2\right)} - \frac{149201 - 214796 L - 1135 L^2}{496726 + 75117 L + 512 L^2} - \frac{79045 - 118430 L - 182 L^2}{599789 + 91562 L + 512 L^2} + \frac{9 \left(25384532 + 113745 L\right)}{64 \left(7149852 + 150141 L + 512 L^2\right)} - \frac{80396 + 135589 L + 798 L^2}{16 \left(38735 + 3105 L + 32 L^2\right)};$$
$$\text{exprint} = \text{MonotoneEnclosure[expr, L, Lint]};$$
$$\text{exprint} + \frac{5939\,\text{Lint}}{13107200000} + \frac{77}{512} \text{Log}[1 + \text{Lint}]\Big];$$

In[170]:=
$$N\left[\text{toprovenonneg}\left[\text{Interval}\left[\left\{2^{12}, \infty\right\}\right]\right], 30\right]$$

Out[170]=
```
Interval[{0.244994201896697407871343431869, ∞}]
```

In[171]:=
```
AbsoluteTiming[
  ProveNonNegative[toprovenonneg, toprovenonneg, {Linterval}, MaxDepth → 25]]
```

Out[171]=
$$\left\{0.043593, \{\}, \{\}, \left\{\left\{\text{Interval}\left[\left\{\frac{1351}{50}, 32\right\}\right]\right\},\right.\right.$$
```
{Interval[{32, 64}]}, {Interval[{64, 128}]}, {Interval[{128, 256}]},
{Interval[{256, 512}]}, {Interval[{512, 1024}]}, {Interval[{1024, ∞}]}}}
```